





Knowledge Engineering with Semantic Web Technologies

Lecture 2: RDF Based Knowledge Representation 2.8 SPARQL Subqueries and Property Paths



Dr. Harald Sack
Hasso-Plattner-Institut for IT Systems Engineering
University of Potsdam
Autumn 2015



SPARQL - Subqueries

- Example: Select all authors, by whom they are influenced and all the influencers' notable works

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX dbo: <http://dbpedia.org/ontology/>

SELECT ?author ?influencer ?work
FROM <http://dbpedia.org/>
WHERE {
  { SELECT ?author ?influencer
    FROM <http://dbpedia.org/>
    WHERE {
      ?author rdf:type dbo:Writer ;
      dbo:influencedBy ?influencer .
    }
  }
  ?influencer dbo:notableWork ?work .
}
LIMIT 100
```



- Subqueries are a way to embed SPARQL queries within other queries
- result is achieved by first evaluating the inner query

[query SPARQL endpoint](#)

SPARQL - Property Paths

- A **property path** is a possible route through an RDF graph between two graph nodes.
 - trivial case: property path of length 1, i.e. a triple pattern
 - **alternatives**: match one or both possibilities

```
{ :book1 dc:title|rdfs:label ?displayString }
```

- **sequence**: property path of length >1

```
{ ?x foaf:mbox <mailto:alice@example> .  
  ?x foaf:knows/foaf:knows/foaf:name ?name . }
```


- **inverse property paths**: reversing the direction of the triple

```
{ ?x foaf:mbox <mailto:alice@example> }  
=  
{ <mailto:alice@example> ^foaf:mbox ?x }
```

SPARQL - Property Paths

- inverse path sequences


```
{ ?x foaf:knows/^foaf:knows ?y  
  FILTER (?x != ?y) . }
```



SPARQL - Property Paths

- arbitrary length match

```
{ ?x foaf:mbox <mailto:alice@example> .  
  ?x foaf:knows+/foaf:name ?name . }
```



SPARQL - Property Paths

- **inverse path sequences**

```
{ ?x foaf:knows/^foaf:knows ?y  
  FILTER (?x != ?y) . }
```

- **arbitrary length match**

```
{ ?x foaf:mbox <mailto:alice@example> .  
  ?x foaf:knows+/foaf:name ?name . }
```

- **negated property paths**

```
{ ?x !(rdf:type|^rdf:type) ?y . }
```

SPARQL - Property Paths

- Example: Who are the authors who were influenced by the influencers of Jules Verne?

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX : <http://dbpedia.org/resource/>
PREFIX dbo: <http://dbpedia.org/ontology/>

SELECT ?influencedByInfluencers
FROM <http://dbpedia.org/>
WHERE {
    :Jules_Verne dbo:influencedBy/^dbo:influencedBy ?influencedByInfluencers
    FILTER (?influencedByInfluencers != :Jules_Verne).
}
```

property path expression

STARQL



EXTRA: 09 - RDF Databases

OpenHPI - Course Knowledge Engineering with Semantic Web Technologies

Lecture 2: RDF Based Knowledge Representation